[bookmark: _GoBack]*&---------------------------------------------------------------------*
*& Report  ZAA_BACKUP_REPORT_TO_TXT.
*&
*&---------------------------------------------------------------------*
*& Description:
*&  Download source code of selected programs and save
*&  them into txt files
*&
*& Technical description:
*&  Program automatically recognizes on what operation system does SAP
*&  system run. According to this information it dynamically creates
*&  OS compatible path.
*&  (Program assumes that SAP system runs on client's PC)
*&
*&---------------------------------------------------------------------*
*& Change log:
*& Date       Author        Action
*& 2017-02-01 ABAP_ACADEMY  Created
*&
*&---------------------------------------------------------------------*

REPORT  zaa_backup_report_to_txt.

*-----------------------------------*
* DECLARATIONS

" Constants

CONSTANTS:
  BEGIN OF gc_download_status,
    success TYPE c VALUE 'S',
    error   TYPE c VALUE 'E',
  END OF gc_download_status.

" Types

TYPES:
  tv_path              TYPE c LENGTH 255, " Download Path
  tv_program_name      TYPE sobj_name,
  tv_source_code_line  TYPE string.

TYPES:
  BEGIN OF ts_log,     " Logging Download Process to This Structure
    object TYPE char4, " Name of the Program
    status TYPE c,     " Status can contain 2 states:
                       " 'S' - Successful Download of Program
                       " 'E' - Error in Download
    ddtext TYPE ddtext, " Long Text Description
  END OF ts_log.

TYPES:
  tt_program_names  TYPE TABLE OF tv_program_name,
  tt_source_code    TYPE TABLE OF tv_source_code_line,
  tt_logs           TYPE TABLE OF ts_log.

" Variables

DATA:
  gv_devclass              TYPE devclass,         " Development Class/Package where programs are stored
  gv_program_name          TYPE tv_program_name,  " Program Name for Download
  gv_folder                TYPE string.           " Target Folder Where Files Will Be Stored

" Internal Tables

DATA:
  gt_program_names  TYPE tt_program_names,        " All Program Names Ready to Be Downloaded
  gt_logs           TYPE tt_logs.                 " Download Process Log


*-----------------------------------*
* SELECTION-SCREEN

" Download Either All Programs from Package or Select Specific Programs
SELECTION-SCREEN BEGIN OF BLOCK b01 WITH FRAME TITLE text-b01.
SELECT-OPTIONS: so_devc FOR gv_devclass.
SELECT-OPTIONS: so_prog FOR gv_program_name.
SELECTION-SCREEN END OF BLOCK b01.

" Choose Download Path to Local Hard Drive
SELECTION-SCREEN BEGIN OF BLOCK b02 WITH FRAME TITLE text-b02.
PARAMETERS: p_path TYPE tv_path OBLIGATORY.
SELECTION-SCREEN END OF BLOCK b02.

" Display Log Information on the Screen
SELECTION-SCREEN BEGIN OF BLOCK b03 WITH FRAME TITLE text-b03.
PARAMETERS: p_succ AS CHECKBOX DEFAULT abap_true.
PARAMETERS: p_err  AS CHECKBOX DEFAULT abap_true.
SELECTION-SCREEN END OF BLOCK b03.


*-----------------------------------*
* AT SELECTION-SCREEN Event Implementation

AT SELECTION-SCREEN ON VALUE-REQUEST FOR p_path.

  " Display Window to Choose Target Path on Local Hard Drive
  CALL METHOD cl_gui_frontend_services=>directory_browse
    CHANGING
      selected_folder = gv_folder
    EXCEPTIONS
      cntl_error      = 1
      error_no_gui    = 2
      OTHERS          = 3.

  IF sy-subrc = 0.
    p_path = gv_folder.
  ENDIF.


*-----------------------------------*
* MAIN LOGIC

START-OF-SELECTION.

  " Check packages or programs are selected for download
  IF so_devc IS INITIAL AND so_prog IS INITIAL.
    WRITE: icon_red_light, 'Firstly choose either development package or program to download!'.
    EXIT. " Program
  ENDIF.

  " Search for programs in the system
  PERFORM get_programs_to_download
    TABLES so_devc
           so_prog
    CHANGING gt_program_names.

  " When no programs found to download, leave the program
  IF gt_program_names IS INITIAL.
    WRITE: /, icon_red_light, 'No programs found for download'.
    EXIT. " Program
  ENDIF.

  " Download programs to local hard drive
  PERFORM download_programs_to_hdd
    USING
      p_path
      gt_program_names
    CHANGING
      gt_logs.

  " Display log on the screen
  PERFORM display_logs
   USING gt_logs
         p_succ
         p_err.

END-OF-SELECTION.


*-----------------------------------*
* SUBROUTINE IMPLEMENTATIONS


*&---------------------------------------------------------------------*
*&      Form  download_reports_to_hdd
*&---------------------------------------------------------------------*
*       Download Source Codes of All Programs and Save Them to HDD
*       Log Results for All Program Downloads
*----------------------------------------------------------------------*
*      -->IV_PATH           (INPUT) Target Path
*      -->IT_PROGRAM_NAMES  (INPUT) Program Names Ready for Download
*      -->ET_LOGS           (OUTPUT)Download Activity Log
*----------------------------------------------------------------------*
FORM download_programs_to_hdd
  USING    iv_path                      TYPE tv_path
           it_program_names             TYPE tt_program_names
  CHANGING et_logs                      TYPE tt_logs.

  DATA:
    lb_is_source_code_get TYPE abap_bool.
  DATA:
    lv_file_path          TYPE string,
    lv_program_name       TYPE tv_program_name.
  DATA:
    ls_log                TYPE ts_log.
  DATA:
    lt_source_code        TYPE TABLE OF tv_source_code_line.

  " Initialize exporting parameters
  CLEAR et_logs.

  " For each program separately do the actions inside of this cycle
  LOOP AT it_program_names INTO lv_program_name.

    " Initialize variable before next cycle starts
    CLEAR:
      lb_is_source_code_get,
      lt_source_code.

    " Get source code of the program
    PERFORM get_source_code
      USING
        lv_program_name
      CHANGING
        lb_is_source_code_get
        lt_source_code
        et_logs.
    IF lb_is_source_code_get = abap_false.
      CONTINUE. " Skip upcomming functionality and go to next cycle
    ENDIF.

    " Create path with target filename where source code will be downloaded
    " E.g. when program name is ZAA_TEST_PROGRAM, final file path will be the following:
    " <IV_PATH - Path to Target Folder><OS Separator><File Name - NSP_ZAA_TEST_PROGRAM.abap>
    PERFORM create_path_to_file
      USING iv_path
            lv_program_name
      CHANGING lv_file_path.

    " Download source code to file on PC
    CALL FUNCTION 'GUI_DOWNLOAD'
      EXPORTING
        filename                = lv_file_path
      TABLES
        data_tab                = lt_source_code
      EXCEPTIONS
        file_write_error        = 1
        no_batch                = 2
        gui_refuse_filetransfer = 3
        invalid_type            = 4
        no_authority            = 5
        unknown_error           = 6
        header_not_allowed      = 7
        separator_not_allowed   = 8
        filesize_not_allowed    = 9
        header_too_long         = 10
        dp_error_create         = 11
        dp_error_send           = 12
        dp_error_write          = 13
        unknown_dp_error        = 14
        access_denied           = 15
        dp_out_of_memory        = 16
        disk_full               = 17
        dp_timeout              = 18
        file_not_found          = 19
        dataprovider_exception  = 20
        control_flush_error     = 21
        OTHERS                  = 22.
    IF sy-subrc <> 0.

      " Add Error Information to Log
      ls_log-status = gc_download_status-error.
      CONCATENATE lv_program_name ' was not downloaded succesfully'
        INTO ls_log-ddtext.
      APPEND ls_log TO et_logs.

    ELSE.

      " Add Error Information to Log
      ls_log-status = gc_download_status-success.
      CONCATENATE lv_program_name ' was downloaded succesfully'
        INTO ls_log-ddtext.
      APPEND ls_log TO et_logs.

    ENDIF.

  ENDLOOP.

ENDFORM.                    " download_reports_to_hdd

*&---------------------------------------------------------------------*
*&      Form  get_programs_to_download
*&---------------------------------------------------------------------*
*       Get all program names from TADIR system table
*----------------------------------------------------------------------*
*      -->IT_DEVC           (INPUT) Development Classes/Packages
*      -->IT_PROG           (INPUT) Program Names
*      -->CT_PROGRAM_NAMES  (OUTPUT)Program Names Ready to Be Downloaded
*----------------------------------------------------------------------*
FORM get_programs_to_download
  TABLES   it_devc
           it_prog
  CHANGING ct_program_names  TYPE tt_program_names.

  " Get all reports from TADIR by provided devclass and programs
  " chosen from selection screen
  SELECT obj_name AS name FROM tadir
    APPENDING TABLE ct_program_names
    WHERE pgmid = 'R3TR'
      AND object = 'PROG'
      AND obj_name IN it_prog
      AND devclass IN it_devc
      AND delflag <> 'X'.

ENDFORM.                    " GET_REPORTS

*&---------------------------------------------------------------------*
*&      Form  get_source_code
*&---------------------------------------------------------------------*
*       Get Source Code of the Requested Program Name
*----------------------------------------------------------------------*
*      -->IV_PROGRAM_NAME        (INPUT) Program Name
*      -->EB_IS_SOURCE_CODE_GET  (OUTPUT)(BOOL) Was this soubroutine
*                                able to get source code?
*      -->ET_SOURCE_CODE         (OUTPUT)Exported Source Code
*      -->CT_LOGS                (OUTPUT)Log Information
*----------------------------------------------------------------------*
FORM get_source_code
  USING
    iv_program_name       TYPE tv_program_name
  CHANGING
    eb_is_source_code_get TYPE abap_bool
    et_source_code        TYPE tt_source_code
    ct_logs               TYPE tt_logs.

  DATA:
    ls_log TYPE ts_log.

  " Initialize exporting parameters
  CLEAR: et_source_code, eb_is_source_code_get.

  " Read source code of report
  READ REPORT iv_program_name INTO et_source_code.
  IF sy-subrc <> 0.

    " We were not able to get source code
    eb_is_source_code_get = abap_false.

    " Add information about problem to log
    ls_log-status = 'R'.
    CONCATENATE iv_program_name ' - problem reading report'
      INTO ls_log-ddtext.
    APPEND ls_log TO ct_logs.

    EXIT. " FORM execution

  ENDIF.

  " We have got source code successfuly
  eb_is_source_code_get = abap_true.

ENDFORM.                    "get_source_code

*&---------------------------------------------------------------------*
*&      Form  create_path_to_file
*&---------------------------------------------------------------------*
*       Create Target Path to File on Local Hard Drive
*----------------------------------------------------------------------*
*      -->IV_PATH          (INPUT) Path
*      -->IV_PROGRAM_NAME  (INPUT) Program Name
*      -->EV_FILE_PATH     (OUTPUT)Target File Path
*----------------------------------------------------------------------*
FORM create_path_to_file
  USING iv_path         TYPE tv_path
        iv_program_name TYPE tv_program_name
  CHANGING ev_file_path TYPE string.

  DATA:
   lv_filename  TYPE string,
   lv_separator TYPE c.

  CLEAR ev_file_path.

  " Create filename for the file containing the source code
  " E.g. when program name is ZAA_TEST_PROGRAM, filename will be called:
  " 'NSP_ZAA_TEST_PROGRAM.abap'
  PERFORM create_filename
    USING iv_program_name
    CHANGING lv_filename.

  " Get folder separator of current operation system
  PERFORM get_os_separator
    CHANGING lv_separator.

  " Create file path, where file will be stored
  CONCATENATE iv_path lv_separator lv_filename
    INTO ev_file_path.

ENDFORM.                    "create_path_to_file

*&---------------------------------------------------------------------*
*&      Form  create_filename
*&---------------------------------------------------------------------*
*       Create Filename of Where Program Will Be Stored
*----------------------------------------------------------------------*
*      -->IV_PROGRAM_NAME  (INPUT) Program Name
*      -->EV_FILENAME      (OUTPUT)Target Filename
*----------------------------------------------------------------------*
FORM create_filename
  USING iv_program_name TYPE tv_program_name
  CHANGING ev_filename  TYPE string.

  DATA:
    lv_program_name TYPE tv_program_name.

  " Initilialize exporting parameter
  CLEAR ev_filename.

  lv_program_name = iv_program_name.

  " In case that program name starts with '/', remove it from the beginning of the filename
  IF iv_program_name(1) = '/'.
    SHIFT lv_program_name LEFT DELETING LEADING '/'.
  ENDIF.

  " Filename Pattern
  " <System ID>_<Program Name>.abap
  CONCATENATE sy-sysid '_' lv_program_name '.abap'
    INTO ev_filename.

  " If program name contains '/' sign, change it to '_'
  REPLACE ALL OCCURRENCES OF '/' IN ev_filename WITH '_'.

ENDFORM.                    "create_filename

*&--------------------------------------------------------------------------------------*
*&      Form  display_logs
*&--------------------------------------------------------------------------------------*
*       Display Activity Log
*---------------------------------------------------------------------------------------*
*      -->IT_LOGS                 (INPUT) Activity Log
*      -->IB_IS_SUCCESS_REQUESTED (INPUT) Is Display of Success Log Messages Requested?
*      -->IB_IS_ERROR_REQUESTED   (INPUT) Is Display of Error Log Messages Requested?
*---------------------------------------------------------------------------------------*
FORM display_logs
  USING   it_logs                 TYPE tt_logs
          ib_is_success_requested TYPE abap_bool
          ib_is_error_requested   TYPE abap_bool.

  DATA: ls_log TYPE ts_log.

  " Display Error Messages if Requested
  IF ib_is_error_requested = abap_true.

    LOOP AT it_logs INTO ls_log WHERE status = gc_download_status-error.
      WRITE: /, icon_red_light, ls_log-object, ls_log-ddtext.
    ENDLOOP.

  ENDIF.

  " Display Success Messages if Requested
  IF ib_is_success_requested = abap_true.

    LOOP AT it_logs INTO ls_log WHERE status = gc_download_status-success.
      WRITE: /, icon_green_light, ls_log-object, ls_log-ddtext.
    ENDLOOP.

  ENDIF.

ENDFORM.                    " DISPLAY_LOGS

*&---------------------------------------------------------------------*
*&      Form  get_os_separator
*&---------------------------------------------------------------------*
*       Get OS Specific Folder Separator
*----------------------------------------------------------------------*
*      -->EV_SEPARATOR  (OUTPUT) OS Specific Separator
*----------------------------------------------------------------------*
FORM get_os_separator
  CHANGING ev_separator TYPE c.

  CONSTANTS:
    BEGIN OF lc_separator,
      windows TYPE c VALUE '\',
      unix    TYPE c VALUE '/',
    END OF lc_separator.

  CLEAR ev_separator.

  IF sy-opsys CA 'Win'.
    ev_separator = lc_separator-windows.
  ELSE.
    ev_separator = lc_separator-unix.
  ENDIF.

ENDFORM.                    "get_os_separator
